query: #Most frequent occupations without an article in Spanish (es)
  #by Edgars2007/Jura1, 2016-05-09
  SELECT ?occup ?occupLabel ?count
  WHERE
  {
  	{
  		SELECT ?occup (COUNT(?person) as ?count)
  		WHERE
  		{
  			?person wdt:P106 ?occup
  		}
  		GROUP BY ?occup
  		ORDER BY DESC(?count)
  		LIMIT 1000
  	}
  
  	# look for articles (sitelinks) in Spanish ("es")
    	OPTIONAL { ?sitelink schema:about ?occup . ?sitelink schema:inLanguage "es" } 
  	# but select items with no such article
  	FILTER (!BOUND(?sitelink))
    
      SERVICE wikibase:label { bd:serviceParam wikibase:language "es,en,pl,ja,zh,ru" }
  }
  ORDER BY DESC(?count)
  LIMIT 50